


地质绘图程序集

牟中海 蒋裕强 编著 (1)



地质出版社

前 言

随着计算机的普及,计算机辅助专业绘图已成为一种必不可少的手段。特别在地质学领域,人们常常借助计算机完成复杂而繁重的工作,并以图形的方式把地质现象表示出来。然而,尽管关于计算机绘图的基本原理、语言方面的论著不少,但关于如何用计算机来完成这一地质工作的方法、技巧方面的书籍却不多见。笔者正是从这一点出发,编写了这本书,供读者参考。

书中所有程序(除菜单外)采用 FORTRAN77 程序设计语言编写,符合结构化原则,并且采用了缩进规则,方便读者阅读程序。尽管它为绘图程序集,但却未采用一条 FORTRAN 绘图语句,因而读者可以非常方便地将其改写为 BASIC 语言、Windows 系统下的 Visual Basic 和 Visual C++ 语言等。

书中所有程序均在 IBM-PC 系列及其兼容机上调试通过。书中所有子程序均给出了所有形式参数的类型及意义,以方便读者使用。除了基本绘图子程序外,其它程序均给出了实例及程序运行结果图件。

本书所有程序所形成的图形文件为文本文件,绘图指令为 Surfer 的 4 个指令:TR X, Y 为图形原点坐标指令,X 和 Y 为图形原点坐标;SP N 为颜色代号指令,N 为颜色代号(或绘图仪的笔号),取值范围为 1~16;MA X, Y 为抬笔移动指令,X 和 Y 为移动点坐标;PA X, Y 为落笔移动指令,X 和 Y 为划线点坐标。读者可方便地将这 4 个指令修改为您的绘图仪指令或其它计算机语言的绘图命令;也可以不做任何修改,利用 Surfer 的设备更改程序 Install 将图形文件直接转换为 AUTOCAD 格式的图形文件,再利用 AUTOCAD 强大的功能编辑、输出图形。

本书中 1、8、9 部分由牟中海、蒋裕强共同完成,3、4 部分由蒋裕强完成,2、5、6、7 部分由牟中海完成。

书不当之处,敬请读者指正。

西南石油学院

牟中海 蒋裕强

1999.1

目 录

前 言

1 基本绘图子程序	1
1.1 建立图形文件子程序	1
1.2 画直线、点子程序	1
1.3 选笔或颜色子程序	2
1.4 图形、字符旋转子程序	2
1.5 画圆子程序	2
1.6 画椭圆子程序	3
1.7 画抛物线子程序	3
1.8 画渐开线子程序	4
1.9 画箭头子程序	4
1.10 画矩形网格子程序	6
1.11 画、填充矩形子程序	7
1.12 写英文字符串子程序	9
1.13 写数值子程序	10
1.14 写中、英文字符串子程序	11
2 多边形图形裁剪	21
3 画等值线图	28
4 填充剖面线 (画盆地边界)	46
5 曲面图	51
6 地质界线	65
7 岩性柱状图	73
8 字符	85
9 菜单	86
主要参考文献	99

1 基本绘图子程序

1.1 建立图形文件子程序

本书所介绍的图形文件均为矢量图形文件，所以在建立任何图形的时候，必须首先调用该子程序。程序运行结束后建立的图形文件名为 PLTFIL。可用 Surfer 的图形浏览器 View 观看图形，Plot 文件打印图形，Install 配置输出设备。

```
SUBROUTINE PLOTS (X0, Y0, L, PLTFIL)
C      X0: 图形圆点 X 坐标。单位: (cm)。输入参数。
C      Y0: 图形圆点 Y 坐标。单位: (cm)。输入参数。
C      L:  打开、关闭图形文件控制符。输入参数。
C          L=1 为打开图形文件;
C          L=2 为关闭图形文件;
C          L=3 为追加图形文件。
C      PLTFIL: 图形文件名。输入参数。
CHARACTER PLTFIL * 12
IF (L. EQ. 1) THEN
    OPEN (50, FILE=PLTFIL, STATUS='UNKNOWN')
C      图形单位为英寸时: CI=0.39372, 单位为厘米时: CI=1.0。
    CI=0.39372
    WRITE (50, ' (A3, 2F8.3) ') 'TR ', X0*CI, Y0*CI
ELSE IF (L. EQ. 2) THEN
    CLOSE (50)
C      WRITE (*, *) '          图形文件名: ', PLTFIL
ELSE IF (L. EQ. 3) THEN
    OPEN (50, FILE=PLTFIL, ACCESS=" APPEND", STATUS='UNKNOWN')
ENDIF
END
```

1.2 画直线、点子程序

该子程序采用的是 Surfer 的抬笔、落笔指令。根据需要，用户可以很容易地将其改写为 AUTOCAD、各型绘图仪、各种计算机语言的抬笔、落笔（或划线）指令。

```
SUBROUTINE PLOT (X, Y, N)
C      X: 绝对横坐标。单位: (cm)。输入参数。
C      Y: 绝对纵坐标。单位: (cm)。输入参数。
C      N: 抬笔、落笔控制符。N=2 为落笔移; N=3 为抬笔移。输入参数。
CI=0.39372
C      CI=1.0
IF (N. EQ. 3) THEN
    WRITE (50, ' (A3, 2F8.3) ') 'MA ', X*CI, Y*CI
```

```

ELSE IF (N.EQ.2) THEN
  WRITE (50, ' (A3, 2F8.3) ') 'PA ', X*CI, Y*CI
ENDIF
END

```

1.3 选笔或颜色子程序

```

SUBROUTINE COLOR (NO, X1, X2, X3)
C   NO: 笔号或颜色代号。输入参数。
C   X1, X2, X3: 未定义。在用于 Windows 系统的 VB、VC 语言中时取整数, 分别代表
C   红、绿、蓝三原色。输入参数。
C   SP 为 Surfer 的笔号或颜色指令。
  WRITE (50, ' (A3, I3) ') 'SP ', NO
  Y1=X1
  Y2=X2
  Y3=X3
END

```

1.4 图形、字符旋转子程序

```

SUBROUTINE ROTATE (OX, OY, X, Y, ANGLE2)
C   OX: 支点 X 坐标。单位: (cm)。输入参数。
C   OY: 支点 Y 坐标。单位: (cm)。输入参数。
C   X: 任意一点的 X 坐标。单位: (cm)。输入、输出参数。
C   Y: 任意一点的 Y 坐标。单位: (cm)。输入、输出参数。
C   ANGLE2: 旋转角度。单位: 弧度。
  IF (X.EQ.0.0) X=0.000001
  IF (Y.EQ.0.0) Y=0.000001
  IF (X.EQ.OX) X=X+0.000001
  IF (Y.EQ.OY) Y=Y+0.000001
  XY=SQRT ((X-OX)**2+(Y-OY)**2)
  ANG=ACOS ((X-OX)/XY)
  X=XY*COS (ANGLE2+ANG)
  Y=XY*SIN (ANGLE2+ANG)
END

```

1.5 画圆子程序

```

SUBROUTINE CIRCLE (X, Y, R, THICK)
C   X, Y: 圆点 X、Y 坐标。单位: (cm)。输入参数。
C   R: 圆半径。单位: cm。输入参数。
C   THICK: 圆厚度 (>=0.02cm)。单位: (cm)。输入参数。
  D0=20.
  N1=360./D0
  N2=THICK/0.02
  DO 10 I2=1, N2
  DO 20 I1=0, N1
    XX=X+ (R+ (I2-1) * 0.02) * COS (I1 * D0 * 3.14159/180)

```

```

YY=Y+ (R+ (I2-1) * 0.02) * SIN (I1 * D0 * 3.14159/180)
IF (I2.EQ.1.AND.I1.EQ.0) THEN
  CALL PLOT (XX, YY, 3)
ENDIF
CALL PLOT (XX, YY, 2)
20  CONTINUE
10  CONTINUE
END

```

1.6 画椭圆子程序

```

SUBROUTINE ELLIPSE (X0, Y0, WL, WS, A1, A2, A3)
C   X0, Y0: 椭圆中心 X、Y 坐标。单位: (cm)。输入参数。
C   WL, WS: 椭圆长半轴、短半轴长度。单位: (cm)。输入参数。
C   A1: 椭圆长轴与 X 轴夹角。单位: 度。输入参数。
C   A2: 椭圆弧起点离心角。单位: 度。输入参数。
C   A3: 椭圆弧终点离心角。单位: 度。输入参数。
A1=A1 * 0.017453
E=A2 * 0.017453
Z=A3 * 0.017453
CA=COS (A1)
SA=SIN (A1)
D1=COS (E)
D2=SIN (E)
IF (E) 10, 20, 20
20  XS=WL * D1 * CA-WS * D2 * SA+X0
   YS=WL * D1 * SA+WS * D2 * CA+Y0
   GOTO 30
10  XS=WL * D1 * CA+WS * D2 * SA+X0
   YS=WL * D1 * SA-WS * D2 * CA+Y0
30  CALL PLOT (XS, YS, 3)
   DT=2 * 0.017453
   N=INT ( (Z-E) /DT+0.5)
   T=E
   DO I=1, N
     T=T+DT
     XX=WL * COS (T)
     YY=WS * SIN (T)
     XI=XX * CA-YY * SA+X0
     YI=XX * SA+YY * CA+Y0
     CALL PLOT (XI, YI, 2)
   ENDDO
END

```

1.7 画抛物线子程序

```

SUBROUTINE PARABOLIC (X1, Y1, X2, Y2, X3, Y3, N)

```

C X1, Y1: 第一个端点的 X、Y 坐标。单位: (cm)。输入参数。
 C X2, Y2: 第二个端点的 X、Y 坐标。单位: (cm)。输入参数。
 C X3, Y3: 控制抛物线端点斜率的点的 X、Y 坐标。单位: (cm)。输入参数。
 C N: 插值点数。决定所画抛物线的光滑度。输入参数。
 DT=1./FLOAT (N)
 T=0.
 CALL PLOT (X1, Y1, 3)
 DO I=1, N
 T=T+DT
 X= (X2-2*X3+X1) * T * * 2+2* (X3-X1) * T+X1
 Y= (Y2-2*Y3+Y1) * T * * 2+2* (Y3-Y1) * T+Y1
 CALL PLOT (X, Y, 2)
 ENDDO
 END

1.8 画渐开线子程序

SUBROUTINE INVOLUTE (X0, Y0, R, AF1, AF2, N)
 C X0, Y0: 基圆的圆心 X、Y 坐标。单位: (cm)。输入参数。
 C R: 基圆的半径。单位: (cm)。输入参数。
 C AF1: 渐开线的起画点与基圆的圆心连线与 X 轴的夹角。单位: 度。输入参数。
 C AF2: 渐开线的张角。单位: 度。输入参数。
 C N: 插值点数。决定所画渐开线的光滑度。输入参数。
 S=AF1 * 0.017453
 DT= (AF2/N) * 0.017453
 XS=X0+R * COS (S)
 YS=Y0+R * SIN (S)
 CALL PLOT (XS, YS, 3)
 DO I=0, N
 SC=DT * I
 T=S+SC
 X=X0+R * (COS (T) + (T-S) * SIN (T))
 Y=Y0+R * (SIN (T) - (T-S) * COS (T))
 CALL PLOT (X, Y, 2)
 ENDDO
 END

1.9 画箭头子程序

SUBROUTINE ARW (X0, Y0, ANG, ARWL, ARWL0, WIDTH, M)
 C X0, Y0: 箭头的起点 X、Y 坐标。单位: (cm)。输入参数。
 C ANG: 箭柄与 X 轴的夹角。单位: 度。输入参数。
 C ARWL: 箭头加箭柄的长度。单位: (cm)。输入参数。
 C ARWL0: 箭头的长度。单位: (cm)。输入参数。
 C WIDTH: 箭头的宽度。单位: (cm)。输入参数。
 C M: 箭头类型控制符。M=1 为单向双面箭头; M=2 为双向双面箭头;
 C M=3 为单向单面箭头; M=4 为双向单面箭头。输入参数。

```

DIMENSION Q (25)
A=0.0174533 * ANG
X1=ARWL0 * COS (A)
Y1=ARWL0 * SIN (A)
X2=WIDTH/2 * SIN (A)
Y2=WIDTH/2 * COS (A)
XE=X0+ARWL * COS (A)
YE=Y0+ARWL * SIN (A)
Q(1)=X0
Q(2)=Y0
N=2
IF (M.EQ. 1.OR. M.EQ. 3) GOTO 10
Q(3)=X0+X1
Q(4)=Y0+Y1
Q(5)=Q (3) -X2
Q(6)=Q (4) +Y2
Q(7)=X0
Q(8)=Y0
N=8
IF (M.EQ. 4) GOTO 10
Q(9)=X0+X1+X2
Q(10)=Y0+Y1-Y2
Q(11)=Q (3)
Q(12)=Q (4)
N=12
10 Q(N+1) =XE
Q(N+2) =YE
Q(N+3) =XE-X1+X2
Q(N+4) =YE-Y1-Y2
Q(N+5) =XE-X1
Q(N+6) =YE-Y1
N=N+6
IF (M.EQ. 3.OR. M.EQ. 4) GOTO 20
Q(N+1) =Q (N-1) -X2
Q(N+2) =Q (N) +Y2
Q(N+3) =XE
Q(N+4) =YE
N=N+4
20 N=N-2
DO I=1, N, 2
  CALL PLOT (Q (I), Q (I+1), 3)
  CALL PLOT (Q (I+2), Q (I+3), 2)
ENDDO
END

```

1.10 画矩形网格子程序

```
SUBROUTINE GRD (X, Y, DX, DY, NX, NY, ANGLE)
C      X, Y: 矩形网格的左下角 X、Y 坐标。单位: (cm)。输入参数。
C      DX, DY: 在 X、Y 方向上网格线间隔。单位: (cm)。输入参数。
C      NX, NY: 在 X、Y 方向上网格数。输入参数。
C      ANGLE: 网格边与 X 轴的夹角。单位: 度。输入参数。
      ANG=ANGLE * 0.0174533
      X0=X
      Y0=Y
      X1=X+DX * NX * COS (ANG)
      Y1=Y+DX * NX * SIN (ANG)
      E=X1
      F=Y1
      J=NY+1
      DO I=1, J
        A=X0
        B=Y0
        C=X1
        D=Y1
        CALL PLOT (X0, Y0, 3)
        CALL PLOT (X1, Y1, 2)
        X0=C-DY * SIN (ANG)
        Y0=D+DY * COS (ANG)
        X1=A-DY * SIN (ANG)
        Y1=B+DY * COS (ANG)
      ENDDO
      X0=C
      Y0=D
      X1=X
      Y1=Y
      TX=DX * COS (ANG)
      TY=DX * SIN (ANG)
      IF (MOD (NY, 2) .NE. 0) GOTO 15
      TX=-TX
      TY=-TY
      X1=E
      Y1=F
15      J=NX+1
      DO 20 I=1, J
        A=X0
        B=Y0
        C=X1
        D=Y1
        CALL PLOT (X0, Y0, 3)
```

```

CALL PLOT (X1, Y1, 2)
X0=C+TX
Y0=D+TY
X1=A+TX
Y1=B+TY
20  CONTINUE
END

```

1.11 画、填充矩形子程序

```

SUBROUTINE RECTGLE (XLOW, YLOW, XUP, YUP, THICK, IFILL)
C  XLOW, YLOW: 矩形左下角 X、Y 坐标。单位: (cm)。输入参数。
C  XUP, YUP: 矩形右上角 X、Y 坐标。单位: (cm)。输入参数。
C  THICK: 矩形厚度 ( $\geq 0.02\text{cm}$ )。单位: (cm)。输入参数。
C  IFILL: 填充方式控制符。IFILL=1 为不填充; IFILL=2 为全填充;
C      IFILL=3 为用竖线填充; IFILL=4 为用水平线填充;
C      IFILL=5 为用方格网填充; IFILL=6 为用 "X" 填充;
C      IFILL=7 为用 "*" 填充; IFILL=8 为用 "+" 填充;
C      IFILL=9 为用左斜线填充; IFILL=10 为用右斜线填充。
N=THICK/0.02+0.01
T=THICK/N
DO I=0, N-1
    X1= (XLOW-I*T)
    Y1= (YLOW-I*T)
    X2= (XUP+I*T)
    Y2= (YUP+I*T)
    CALL PLOT (X1, Y1, 3)
    CALL PLOT (X2, Y1, 2)
    CALL PLOT (X2, Y2, 2)
    CALL PLOT (X1, Y2, 2)
    CALL PLOT (X1, Y1, 2)
ENDDO
IF (IFILL.EQ.2) THEN
    N= (XUP-XLOW) /0.02+0.01
    CALL PLOT (XLOW, YLOW, 3)
    CALL PLOT (XLOW, YUP, 2)
    DO I=1, N
        CALL PLOT (XLOW+I*0.02, YLOW, 3)
        CALL PLOT (XLOW+I*0.02, YUP, 2)
    ENDDO
ELSE IF (IFILL.EQ.3) THEN
    N= (XUP-XLOW) /0.1+0.01
    DO I=1, N
        CALL PLOT (XLOW+I*0.1, YLOW, 3)
        CALL PLOT (XLOW+I*0.1, YUP, 2)
    ENDDO

```

```

ELSE IF (IFILL.EQ.4) THEN
  N= (YUP-YLOW) /0.1+0.01
  DO I=1, N
    CALL PLOT (XLOW, YLOW+I*0.1, 3)
    CALL PLOT (XUP, YLOW+I*0.1, 2)
  ENDDO
ELSE IF (IFILL.EQ.5) THEN
  N= (XUP-XLOW) /0.1+0.01
  DO I=1, N
    CALL PLOT (XLOW+I*0.1, YLOW, 3)
    CALL PLOT (XLOW+I*0.1, YUP, 2)
  ENDDO
  N= (YUP-YLOW) /0.1+0.01
  DO I=1, N
    CALL PLOT (XLOW, YLOW+I*0.1, 3)
    CALL PLOT (XUP, YLOW+I*0.1, 2)
  ENDDO
ELSE IF (IFILL.EQ.6) THEN
  CALL PLOT (XLOW, YLOW, 3)
  CALL PLOT (XUP, YUP, 2)
  CALL PLOT (XLOW, YUP, 3)
  CALL PLOT (XUP, YLOW, 2)
ELSE IF (IFILL.EQ.7) THEN
  CALL PLOT (XLOW, YLOW, 3)
  CALL PLOT (XUP, YUP, 2)
  CALL PLOT (XLOW, YUP, 3)
  CALL PLOT (XUP, YLOW, 2)
  CALL PLOT (XLOW+ (XUP-XLOW) /2, YLOW, 3)
  CALL PLOT (XLOW+ (XUP-XLOW) /2, YUP, 2)
  CALL PLOT (XLOW, YLOW+ (YUP-YLOW) /2, 3)
  CALL PLOT (XUP, YLOW+ (YUP-YLOW) /2, 2)
ELSE IF (IFILL.EQ.8) THEN
  CALL PLOT (XLOW+ (XUP-XLOW) /2, YLOW, 3)
  CALL PLOT (XLOW+ (XUP-XLOW) /2, YUP, 2)
  CALL PLOT (XLOW, YLOW+ (YUP-YLOW) /2, 3)
  CALL PLOT (XUP, YLOW+ (YUP-YLOW) /2, 2)
ELSE IF (IFILL.EQ.9) THEN
  CALL PLOT (XLOW, YLOW, 3)
  CALL PLOT (XUP, YUP, 2)
ELSE IF (IFILL.EQ.10) THEN
  CALL PLOT (XLOW, YUP, 3)
  CALL PLOT (XUP, YLOW, 2)
ENDIF
END

```

1.12 写英文字符串子程序

关于英文字符库 (ENGLISH.FON) 的详细内容见本书最后一页。

```
SUBROUTINE NUMBERS (OX, OY, SG, HY, HX, A1, A2, JC1)
C      写英文字符串子程序
C      OX, OY: 字符串起点 X、Y 坐标。单位: (cm)。输入参数。
C      SG: 字符串 (<=120个西文字符)。输入参数。
C      HY, HX: 字符高度、宽度。单位: (cm)。输入参数。
C      A1: 字符串倾斜角度。单位: 度。输入参数。
C      A2: 字符串旋转角度。单位: 度。输入参数。
C      JC1: 字颜色代号。输入参数。
CHARACTER SG * 120
ANG2=A2
ANG1=A1
CALL COLOR (JC1, 1., 1., 1.)
ANG1=ANG1 * 3.141592/180.0
ANG2=ANG2 * 3.141592/180.0
DO 40 I=120, 1, -1
    IF (SG (I: I) .GT. '') THEN
        LSTR=I
        GOTO 50
    ENDIF
40  CONTINUE
50  XO = OX
    YO = OY
    II=1
110 IF (II.GT.LSTR) GOTO 120
    ICODE=ICHAR (SG (II: II))
    IF (SG (II: II) .EQ. '') GOTO 101
    CALL SYMENG (XO, YO, SG (II: II), HY, HX, A1, A2, 2)
101 II=II+1
    XO=XO+HX * COS (ANG2)
    YO=YO+HX * SIN (ANG2)
    GOTO 110
120 END

SUBROUTINE SYMENG (XO, YO, SYM, HY, HX, ANGLE1, ANGLE2, NF)
C      XO, YO: 字符起点 X、Y 坐标。单位: (cm)。输入参数。
C      SYM: 字符。输入参数。
C      HY, HX: 字符高度、宽度。单位: (cm)。输入参数。
C      ANGLE1: 字符倾斜角度。单位: 度。输入参数。
C      ANGLE2: 字符旋转角度。单位: 度。输入参数。
C      NF: 字体控制符 (1, 2, 3, 4)。
PARAMETER      (NPMAX = 95, MAXY = 2 * NPMAX)
CHARACTER      BUF (MAXY), SYM
```

```

LOGICAL ENDP
NPOINTS=63
NFONT=NF-1
IF (MOD (NPOINTS, 2) .NE. 0) NPOINTS = NPOINTS + 1
NXYS=2 * NPOINTS
ANG2=ANGLE2
ANG1=ANGLE1
ANG1=ANG1 * 3.141592/180.0
ANG2=ANG2 * 3.141592/180.0
L=NXYS
NCHARS=95
OPEN (3, FILE='ENGLISH.FON', STATUS='OLD', ACCESS='DIRECT', RECL=L)
IASC=ICHAR (SYM)
IREC= (NFONT * NCHARS) + IASC - 32
ENDP=.FALSE.
READ (3, REC=IREC) (BUF (I), I=1, NXYS)
J = 1
DO 30 I = 1, NPMAX
  IX = ICHAR (BUF (J))
  IY = ICHAR (BUF (J+1))
  IF (IX.GT.127) THEN
    IX = IX - 128
    IST = 3
  ELSE
    IST = 2
  ENDIF
  IF (IY.GT.127) THEN
    IY = IY - 128
    ENDP=.TRUE.
  ENDIF
  X=IX * HX * 0.01
  Y=IY * HY * 0.01
  X=X+Y * COS (ANG1)
  CALL ROTATE (0.0, 0.0, X, Y, ANG2)
  X=X+XO
  Y=Y+YO
  IF (IST.EQ.3) CALL PLOT (X, Y, 3)
  IF (IST.EQ.2) CALL PLOT (X, Y, 2)
  IF (ENDP) GOTO 40
30  J=J+2
40  CLOSE (3)
END

```

1.13 写数值子程序

首先用下面的子程序将数值转换为字符串，再调用前述的字符串子程序即可。

SUBROUTINE VALUEconvertSTRING (VALUE, TEXT, NBITS, NUMDIGI)

```

C      将数值转换为字符串子程序
C      VALUE: 数值。输入参数。
C      TEXT: 字符串。输入参数。
C      NBITS: 小数位数。输入参数。
C      NUMDIGI: 转换后的字符个数。输出参数。
      CHARACTER STRING * 20, TEXT * 120
      WRITE (STRING, ' (F11.5, A1) ') VALUE, "C
      DO 1 I=1, 20
      IF (STRING (I: I) .NE.' ') THEN
          L1=I
          GOTO 2
      ENDIF
1      CONTINUE
2      DO 3 I=1, 20
      IF (STRING (I: I) .EQ.' ') THEN
          L2=I
          GOTO 4
      ENDIF
3      CONTINUE
4      IF (NBITS.EQ.0) THEN
          TEXT=STRING (L1: L2-1)
          NUMDIGI=L2-L1
      ELSE
          TEXT=STRING (L1: L2+NBITS)
          NUMDIGI=L2+NBITS-L1+1
      ENDIF
C      将字符串转换为数值语句
C      A='23'
C      B='-123.5'
C      READ (A, ' (I2) ') N
C      READ (B, ' (F7.2) ') E
C      WRITE (*, *) N, E
      END

```

1.14 写中、英文字符串子程序

关于中、英文字符库 (HZK-S.LIB) 的详细内容见本书最后一页。

SUBROUTINE SYMBOLM (OX, OY, SG, HY, HX, DI, A1, A2, KX, KY, JC1, JC2)

```

C      OX, OY: 字符串起点 X、Y 坐标。单位: (cm)。输入参数。
C      SG: 字符串 (<=120个西文字符)。输入参数。
C      HY, HX: 字符串高度、宽度。单位: (cm)。输入参数。
C      DI: 字符间隔。单位: (cm)。输入参数。
C      A1: 字符串倾斜角度。单位: 度。输入参数。
C      A2: 字符串旋转角度。单位: 度。输入参数。
C      KX: 字体控制符。输入参数。

```

```

C      KX=1为楷书；KX=2为宋体；KX=3为仿宋体；KX=4为黑体。
C      KY：字形控制符.输入参数。
C      KY=1为实心字；KY=2为空心字；KY=3为实心加边字；KY=4为立体字。
C      JC1：实心字颜色代号（1~16）。输入参数。
C      JC2：字边缘颜色代号（1~16）。输入参数。
C      设置字库点阵数 MM=?
      PARAMETER (MM=24)
      CHARACTER SG * 120, HH * (MM/8), STR (120) * 2, FILENAME * 12
      DIMENSION IA (0: MM+1, 0: MM+1), L (MM/8)
      ANG2=A2
      ANG1=A1
      ANG1=ANG1 * 3.141592/180.0
      ANG2=ANG2 * 3.141592/180.0
      XO = OX
      YO = OY
      XO1 = OX
      YO1 = OY
C      计算字符高度为 HY、字符宽度为 HX 时每个点的放大倍数 XM、YM
      XM=HX/ (MM * 1.0)
      YM=HY/ (MM * 1.0)
      K2=0
      J=1
      K=1
      DO I=120, 1, -1
        IF (SG (I; I) .NE. " ") THEN
          LL=I
          GOTO 1
        ENDIF
      ENDDO
1      DO I=1, LL
        IF (ICHAR (SG (K; K)) .LT. 127) THEN
          STR (J) =SG (K; K)
          J=J+1
          K=K+1
        ELSE
          STR (J) =SG (K; K+1)
          J=J+1
          K=K+2
          K2=K2+1
        ENDIF
      ENDDO
C      LM：汉字与西文字的个数
      LM=LL-K2
      IF (KX.EQ.1) THEN
C      FILENAME=" HZK-K.LIB"

```

```

ELSE IF (KX.EQ.2) THEN
    FILENAME=" HZK-S.LIB"
ELSE IF (KX.EQ.3) THEN
C    FILENAME=" HZK-FS.LIB"
ELSE IF (KX.EQ.4) THEN
C    FILENAME=" HZK-H.LIB"
ENDIF
OPEN (1, FILE=FILENAME, RECL=MM/8, FORM=" BINARY", ACCESS="
DIRECT")
DO 19 I=1, LM
C    计算该字库01-87区记录号
    IF (STR (I) .EQ." ") THEN
        IC1=32
    ELSE
        IC1=ICHAR (STR (I) (1: 1))
    ENDIF
    IF (IC1.LT.127.AND.IC1.GE.32) THEN
C        第十五区: ASC 码在33-126之间的字符 (即键盘字符)
        LREC= (IC1-32) * MM+94 * 14 * MM-MM+1
        JJ0=MM/2
    ELSE
        LREC= ( (IC1-161) * 94+ICHAR (STR (I) (2: 2)) -161) * MM+1
        JJ0=MM
    ENDIF
    IF (LREC.LE.0) GOTO 19
    DO 20 I1=1, MM
        READ (1, REC=LREC+I1-1) HH
        DO IW=1, MM/8
            L(IW) =ICHAR (HH (IW: IW))
        ENDDO
C        HH (1: 1), L (1):          汉字第 I1行左部 ASC 码字元及对应 ASC 值
C        HH (2: 2), L (2):
C        HH (3: 3), L (3):
C        HH (MM/8: MM/8), L (MM/8): 汉字第 I1行右部 ASC 码字元及对应 ASC 值
C        A (I1, J1): 为汉字字模
C        将 L (MM/8) 转换为二进制数
        DO 50 IW=1, MM/8
            DO J1=8 * IW, 8 * IW-7, -1
                J=MOD (L (IW), 2)
                IF (J.EQ.0) THEN
                    IA (I1, J1) =0
                ELSE
                    IA (I1, J1) =1
                ENDIF
            L (IW) =L (IW) /2

```

```

        ENDDO
50      CONTINUE
C      对于高点阵字，由于数组太大，只能写实心字。这时将 IA (0: MM+1, 0: MM+1) 改为
C      IA (0: MM+1)，并将上下程序的 IA 数组中的 I1和 I2删掉。
C      IF (KY.EQ.1) THEN
C          实心
C          CALL COLOR (JC1, 0.1, 0.1, 0.1)
C          DO 40 I3=0, JJ0+1
C              IF (IA (I3) .NE. 0. AND. IA (I3-1) .EQ. 0) THEN
C                  K1=I3
C              ENDIF
C              IF (IA (I3) .NE. 0. AND. IA (I3+1) .EQ. 0) THEN
C                  K2=I3
C                  X1=XO+ (K1-1) * XM
C                  Y1=YO+HY-I1 * YM
C                  X1=X1+ (Y1-YO) * COS (ANG1)
C                  X2=X1+ (K2-K1+1) * XM
C                  Y2=Y1+YM
C                  XX1=XO
C                  YY1=YO
C                  XX2=X1
C                  YY2=Y1
C                  XX3=X2
C                  YY3=Y2
C                  XX4=ANG1
C                  YY4=ANG2
C                  CALL OBLIQUE (XX1, YY1, XX2, YY2, XX3, YY3, XX4, YY4)
C                  CALL FILLOBRECT (XX1, YY1, XX2, YY2, XX3, YY3, XX4, YY4)
C              ENDIF
C31      CONTINUE
C      ENDIF
20      CONTINUE
C      XO=XO+ (XM * JJ0+DI) * COS (ANG2)
C      YO=YO+ (XM * JJ0+DI) * SIN (ANG2)
C      GOTO 19
C      IF (KY.EQ.1.OR.KY.EQ.3.OR.KY.EQ.4) THEN
C          实心
C          CALL COLOR (JC1, 0.1, 0.1, 0.1)
C          DO 30 I2=1, MM
C              DO 31 I3=0, JJ0+1
C                  IF (IA (I2, I3) .NE. 0. AND. IA (I2, I3-1) .EQ. 0) THEN
C                      K1=I3
C                  ENDIF
C                  IF (IA (I2, I3) .NE. 0. AND. IA (I2, I3+1) .EQ. 0) THEN
C                      K2=I3

```